Kinematic car

Let us consider the following nonlinear ordinary differential system:

```
eqs = {x'[t] → u₁[t] Cos[θ[t]], y'[t] → u₁[t] Sin[θ[t]], θ'[t] → u₂[t]};
vars = {x[t], y[t], θ[t], u₁[t], u₂[t]};
TableForm[eqs]
```

$x'[t] \to \mathrm{Cos}[\theta[t]]\, u_1[t]$
$y'[t] \to \mathrm{Sin}[\theta[t]]\, u_1[t]$
$\theta'[t] \to u_2[t]$

Let us introduce the Ore algebra A defined by

```
replA = ModelToReplacementRules[eqs, t];
A = OreAlgebraWithRelations[Der[t], replA]
```

$\mathbb{K}(t)[D_t; 1, D_t]$

and the matrix R of ordinary differential operators, which defines the generic linearization of the above nonlinear system

```
MatrixForm[R = ToOrePolynomialD[eqs, vars, A]]
```

$$\begin{pmatrix} D_t & 0 & \mathrm{Sin}[\theta[t]]\, u_1[t] & -\mathrm{Cos}[\theta[t]] & 0 \\ 0 & D_t & -\mathrm{Cos}[\theta[t]]\, u_1[t] & -\mathrm{Sin}[\theta[t]] & 0 \\ 0 & 0 & D_t & 0 & -1 \end{pmatrix}$$

and the left A-module, finitely presented by R, defined by M = $A^{1\times 5} / A^{1\times 3} R$.

Let us compute the adjoint of R:

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} -D_t & 0 & 0 \\ 0 & -D_t & 0 \\ \mathrm{Sin}[\theta[t]]\, u_1[t] & -\mathrm{Cos}[\theta[t]]\, u_1[t] & -D_t \\ -\mathrm{Cos}[\theta[t]] & -\mathrm{Sin}[\theta[t]] & 0 \\ 0 & 0 & -1 \end{pmatrix}$$

Let us check whether or not M is torsion-free:

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

$$\left\{ \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & D_t & 0 & -1 \\ 0 & -D_t & \mathrm{Cos}[\theta[t]]\, u_1[t] & \mathrm{Sin}[\theta[t]] & 0 \\ D_t & 0 & \mathrm{Sin}[\theta[t]]\, u_1[t] & -\mathrm{Cos}[\theta[t]] & 0 \end{pmatrix}, \right.$$

$$\left. \begin{pmatrix} -\mathrm{Cos}[\theta[t]]\, u_1[t] & -\mathrm{Sin}[\theta[t]]\, u_1[t]^2 \\ -\mathrm{Sin}[\theta[t]]\, u_1[t] & \mathrm{Cos}[\theta[t]]\, u_1[t]^2 \\ -u_2[t] & u_1[t]\, D_t + 2\, u_1'[t] \\ -u_1[t]\, D_t - u_1'[t] & -u_1[t]^2\, u_2[t] \\ -u_2[t]\, D_t - u_2'[t] & u_1[t]\, D_t^2 + 3\, u_1'[t]\, D_t + 2\, u_1''[t] \end{pmatrix} \right\}$$

Since the first matrix is identity, M is torsion-free, which proves that the generic linearization is controllable. Let us now check whether or not M is free, i.e. that generic linearization is flat.

```
T = LeftInverse[Q, A]
```

$$\left\{\left\{-\frac{\text{Cos}[\theta[t]]}{u_1[t]}, -\frac{\text{Sin}[\theta[t]]}{u_1[t]}, 0, 0, 0\right\}, \left\{-\frac{\text{Sin}[\theta[t]]}{u_1[t]^2}, \frac{\text{Cos}[\theta[t]]}{u_1[t]^2}, 0, 0, 0\right\}\right\}$$

Since Q admits a left inverse T, we conclude that M is a free left A-module of rank 2. In particular, a basis of M, i.e. a flat output
$\xi = (\xi_1, \xi_2)$ of the generic linearization is defined by

```
ApplyMatrix[T, vars]
```

$$\left\{-\frac{1}{u_1[t]} (\text{Cos}[\theta[t]]\, x[t] + \text{Sin}[\theta[t]]\, y[t]), \frac{1}{u_1[t]^2} (-\text{Sin}[\theta[t]]\, x[t] + \text{Cos}[\theta[t]]\, y[t])\right\}$$

Moreover, a parametrization of the flat system is defined by

```
Thread[vars -> ApplyMatrix[Q, {ξ1[t], ξ2[t]}]]
```

$$\{x[t] \to -u_1[t] (\text{Cos}[\theta[t]]\, \xi_1[t] + \text{Sin}[\theta[t]]\, u_1[t]\, \xi_2[t]),$$
$$y[t] \to u_1[t] (-\text{Sin}[\theta[t]]\, \xi_1[t] + \text{Cos}[\theta[t]]\, u_1[t]\, \xi_2[t]),$$
$$\theta[t] \to -u_2[t]\, \xi_1[t] + 2\, \xi_2[t]\, u_1'[t] + u_1[t]\, \xi_2'[t],$$
$$u_1[t] \to -u_1[t]^2\, u_2[t]\, \xi_2[t] - \xi_1[t]\, u_1'[t] - u_1[t]\, \xi_1'[t],$$
$$u_2[t] \to -\xi_1[t]\, u_2'[t] - u_2[t]\, \xi_1'[t] + 3\, u_1'[t]\, \xi_2'[t] + 2\, \xi_2[t]\, u_1''[t] + u_1[t]\, \xi_2''[t]\}$$

From this point we will use some procedures which are not freely available (see NLControl website http://www.nlcontrol.ioc.ee).
Let us check if the flat output of the generic linearization can be simply lifted to flat outputs of the nonlinear system. To do that we consider the one-forms corresponding to the flat outputs of the generic linearization. We try to integrate them.

```
BookForm[difs = ApplyMatrixD[T, {x[t], y[t], θ[t], u1[t], u2[t]}]]
```

$$\left\{-\frac{\text{Cos}[\theta[t]]}{u_1[t]} dx[t] - \frac{\text{Sin}[\theta[t]]}{u_1[t]} dy[t], \frac{\text{Cos}[\theta[t]]}{u_1[t]^2} dy[t] - \frac{\text{Sin}[\theta[t]]}{u_1[t]^2} dx[t]\right\}$$

```
BookForm[sp = SimplifyBasis[SpanK[difs, t]]]
```

SpanK[dx[t], dy[t]]

```
IntegrateOneForms[sp]
```

{x[t], y[t]}

Finally we obtain that x and y define a flat output of the nonlinear system.